Computational Part

Figure 1 compares the effects of a 0.25 shock of monetary policy and investment to output, investment, consumption, q, inflation. Notice that from the picture it may seem that the effect of the shocks vanishes after two periods, but this is because the big shocks at the beginning affect the scale of the picture.

Monetary Shock

The monetary shock lasts less periods (the autocorrelation coefficient is lower) and it causes a big initial drop in output, consumption, investment, q and inflation that almost immediately recovers, becoming positive for some periods before returning to the steady state (consumption returns slowly to the steady state because of consumption smoothing). Intuitively, what happens after a monetary tightening is the following: as monetary policy is non neutral in the NK framework, after the shock the nominal interest rate increases which make consumption decline. Inflation decreases, real wages cannot fully change to the new equilibrium, so mark-up increases, labor supply goes down and similarly output, investment and q drop. Note that investment and consumption depend on the expected future values (mark-up decreases and this makes the system return to the steady state) so they drive the recovery, that is pretty fast under this specification - variables even overshoot - as investment has to be positive for some periods in order to make capital recover to the steady state value. In other words, the presence of investment makes the recovery faster because it overshoots so that the household can return to the same level of consumption as fast as possible.

Investment Shock

An investment shock of the same size of the monetary one, instead, despite being more persistent, has a smaller absolute effect compared to the monetary. As expected the positive investment shock has a positive effect on all the variables (from the pictures it seems zero but it's because of the magnitude of the monetary shock - so I add the investment shock separately in Figure 2).

Figure 1: Monetary and Investment shock, $\varepsilon_{vt} = 0.25$, $\varepsilon_{\chi t} = 0.25$



Figure 2: Investment shock $\varepsilon_{\chi t} = 0.25$



Dynare Code

```
var y c inv l rn pi k q z mu x v; varexo ex ev;

parameters alpha phi little_c theta beta phi_pi
mu_ss delta rho_v rho_x r_ss share_c share_inv y_over_k ni omega;
alpha=1/3;
phi=1;
little_c=2;
theta=0.9;
beta=0.98;
phi_pi=1.1;
mu_ss=0.1;
delta=0.025;
rho_v=0.75;
rho_x=0.95;
y_over_k=(beta^(-1)+delta-1)/alpha;
share_inv=delta*(y_over_k)^(-1);
share_c=1-share_inv;
ni=(1-delta)*((alpha*y_over_k)*(1+mu_ss)^(-1)+1-delta)^(-1);
```

```
omega=(1-theta)*(1-theta*beta)*(theta)^(-1);
r_ss=(beta)^(-1)-1;
model;
//AD
y = share_c*c + share_inv*inv;
c = -(rn-pi(+1))+c(+1);
inv-k(-1) = (delta*little_c)^(-1)*q + x;
q = ((1-ni)*z(+1)-(rn-pi(+1))+ni*q(+1));
z = y-k(-1)-mu;

//AS
y = alpha*k + (1-alpha)*l;
y-l = mu + phi*l + c;
pi = omega*(-mu) + beta*pi(+1);
k = delta*inv + (1-delta)*k(-1);

//MP
rn = phi_pi*pi+v;

//SHOCKS
x = rho_x*x(-1)+ex;
v = rho_v*v(-1)+ev;
end;

initval;
y = 0;
c = 0;
inv = 0;
l = 0;
rn = 0;
pi = 0;
k = 0;
q=0;
z=0;
mu=0;
x=0;
v=0;
end;

shocks;
var ev=0.25^2; //ex=0.25^2;
end;
steady;
stoch_simul(irf=50,order=1);
```